
ABSTRACT
Data quality in research is important. It may be necessary for data from a device to be used
in a research project. Often it is read manually from an external file and entered onto a CRF.
Then the data is manually read from the CRF and entered it into a database. This process
introduces many opportunities for data quality to be compromised. The quality of device data
used in a study can be greatly improved if the data can be read directly from a device’s output
file directly into a dataset. If the device outputs results into a file that can be saved
electronically, SAS® can be used to read the data needed from the results and save the data
directly into a dataset.
Quite often, device data is saved in separate files per subject and it is often difficult to import
each separate file into SAS without great effort. If data is organized with the subject ID as
the folder name and each subject's data in the corresponding folder, SAS® can also be used
to read the data from a general location, importing all data within each location. In addition
to improving data quality, data collection and monitoring time can also be reduced by taking
advantage of these electronic files as opposed to recapturing this data on a CRF.

INTRODUCTION
The methods introduced will allow a SAS® Programmer, with basic SAS® programming skills,
use SAS® to use SAS® to import individual subject data files saved with similar files name and
folder name structures and import desired test results from a report in Excel® or fixed text
format generated by a testing device.

SUBJECT DATA
Very often, data files or test results, generated by a device (MRI reads, Catheterization and
other types of blood testing), are created and saved in separate directories with similar names
per subject and similarly named data files per subject. All this data usually needs to be
imported into one dataset.

Figure 1. Example of Subject Data Directory

IMPORT LIST OF SUBJECTS
When data is organized such that there is one file per subject in a directory, and there is no
set list of subjects, it is simple to obtain a listing of subjects to use for importing all data.

USE A FILENAME STATEMENT TO CREATE THE LIST OF SUBJECTS
A filename statement using pipe and a dir command can be used with an infile dataset to
obtain a listing of the folder names from the desired directory. Using a folder naming
convention that includes the subject’s identifier allows a user to isolate each identifier and
create a listing of all identifiers from the directory in one dataset.

filename dir pipe 'dir "<Source Data Location>" ';
data dirlistxx;
infile dir lrecl=32727 truncover scanover;
input dirtext $200.;
run;

Figure 2. dirlistxx

SUBSET SUBJECT IDS TO IMPORT DATA
Once the dataset with the folder names has been obtained, the subset of the data is created
to exclude any observations that do not contain subject identifiers. A substring of each folder
name isolating each identifier can be created in a separate variable. Only this new variable
is retained. The count of subject identifiers is retained as a macro variable for the loop in the
macro that imports the data.

data dirlist (drop=dirtext);
length subjid $500.;
set dirlistxx;
if dirtext^='' and scan(dirtext,1,'')^='Volume' and
scan(dirtext,1,'')^='Directory' and scan(dirtext,2,'')
^in('File(s)','Dir(s)') and scan(dirtext,-1,'') ^in("..",".");
subjid=substr(scan(dirtext,-1,''),8);
run;
data dirlist;
set dirlistx nobs=subjcnt;
call symput('subjcnt',put(subjcnt,best12.));
run;
proc sort data=dirlist; by subjid; run;

Figure 3. dirlist

IMPORT SUBJECTS’ DATA INTO ONE DATASET
USE A FILENAME STATEMENT TO CREATE THE LIST OF DATA FILES
Similar to above, a listing of results data filenames can be created from each subject’s folder
to use for importing the data into one dataset. A filename statement using pipe and a dir
command can be used with an infile dataset so create a listing of all files in each subject’s
folder. A dataset of this list with only filenames is then retained.

%macro Panel(obsnum=);
%do i=1 %to &obsnum;
data _null_;
set dirlist end=eof;
if _n_=&i then do;
call symput('subjid',strip(trim(subjid)));
end;
run;
%let subjData=&Source.\Subject&subjid;
libname subjData "&subjData.";
%let subjData=%sysfunc(quote(%qsysfunc(dequote
(&subjData))));
filename sdir&i pipe %sysfunc(quote(dir &subjData));
data subjdirXX&i;
infile sdir&i lrecl=32727 truncover scanover;
input subjdir $200.;
run;
data subjdirX&i (keep=panel);
set subjdirXX&i;
if subjdir^='' and scan(subjdir,1,'')^='Volume' and
scan(subjdir,1,'')^='Directory' and scan(subjdir,2,'')
^in('File(s)','Dir(s)') and scan(subjdir,-1,'')
^in('.','..');
panel=cat(scan(subjdir,-3,'. '),'',scan(subjdir,-2,'. '));
run;
data subjdir&i;
set subjdirX&i nobs=panelcnt;
call symput('panelcnt',put(panelcnt,best12.));
run;

Figure 4. subjdir1 (data files for first subject)

EXTRACTING DATA FROM EXCEL® SPREADSHEET REPORTS
GENERATED BY A DEVICE
IMPORT DATA INTO ONE DATASET
A do loop is used to loop through each subject id to create a dataset of each subject’s files to
be imported by storing the id in a macro variable (as seen above). Once the subject’s id is
defined in the do loop, a second do loop is executed to create a dataset containing filenames
of files to be imported in the final dataset.
%do j=1 %to &panelcnt;
data _null_;
set subjdir&i;
if _n_=&j then do;
call symput('panel',trim(strip(panel)));
call
symput('comppanel',trim(strip(compress(panel))));
end;
run;
proc import out=subject&subjid.&comppanel.x
datafile="&Source.\Subject&subjid.\&panel..xlsx"
dbms=xlsx replace;
sheet="&panel";
getnames=yes;
run;
data subject&subjid.&comppanel;
set subject&subjid.&comppanel.x;
subjid="&subjid";
panel="&panel";
run;
data panels;
set panels
subject&subjid.&comppanel;
if subjid^='';
run;
%end;
%end;
%mend;
The structure of the final dataset is determined per the structure of the data in the imported
files and once do loops are executed for all ids, a final dataset that contains all results from
the imported files is created. In this example all files are of XLSX format and panel number
and subject identifier are included in the final dataset.

Figure 5. Source data structure
Execute the macro:

%Panel(obsnum=&subjcnt);

Figure 6. Result of macro execution (All panel data from all subjects)

EXTRACTING DATA FROM TEXT REPORTS GENERATED BY A
DEVICE
DEVICE DATA AS A TEXT FILE
Some devices generate text files containing results from testing. Each file is usually presented
in a fixed format with one file per subject per test.

Figure 7. Example of Device data text file
The results may contain several results for a subject grouped by a particular characteristic.

Figure 8. Example of Device data text file

USING SAS® TO EXTRACT DESIRED DATA
The process of manually entering this data onto a CRF, then into a database can be eliminated
if SAS® is used to extract desired data directly from the text report.
It is best to use this program in tandem with the macro above to obtain the data filenames,
since data for multiple subjects will most likely be imported.

Extraction from a simple report
Extracting data from a report as seen in Figure 8 involves reading the txt file in a data step
using an infile statement, starting at line 3 (where the data starts). For this extraction, only
the testing date (Generated), Patient Initials, Subject ID (Patient ID), Test Heart Rate (Heart
rate), Test Result 6, and Test Result 7 are needed for the final dataset. The subject ID is
assigned to a macro variable to serve as an identifier for later use in the macro. The data
step finds the line containing the desired data value, reads in the value and units and stores
it in a character variable.
%macro Tst2(filename);
data Tst2x;
infile "(Data Location)\&filename..txt" firstobs=3 truncover
scanover;
input
@'Generated:' itstdt $100.
@'Patient Initials:' iinit $100.
@'Patient ID:' isubjid $100.
@'Heart rate:' itst2hr $100.
@'Test Result 6:' itstrslt6 $100.
@'Test Result 7:' itstrslt7 $100.;
call symput('subjid',strip(trim(isubjid)));
run;
The next data step in the macro creates a subject specific dataset that isolates the numeric
values from each character test result variable from the dataset above (dropping the character
variables).

data tst2&subjid. (drop=iinit itstdt isubjid itstrslt6 itstrslt7 itst2hr);
retain subjid;
length subjid $25. init $3.;
format tstdt date9.;
set tst2x;
tstdt=input(scan(itstdt,1,' '),mmddyy10.);
subjid=strip(trim(isubjid));
init=strip(trim(iinit));
tst2hr=input(scan(itst2hr,1,' '),best12.);
tst2rslt6=input(scan(itstrslt6,1,' '),best12.);
tst2rslt7=input(scan(itstrslt7,1,' '),best12.);
run;
proc sort data=tst2&subjid; by subjid tstdt; run;
%mend;
Call the macro with the filename as the parameter:
%Tst2(Subj 0101 LB Test 2);
Resulting in the following output:

Figure 9. Test 2 Result

Extraction from a grouped report
Extracting data from a report as seen in Figure 9 is similar to the above extraction except
data must be extracted per section. The data values needed for the final dataset are the
testing date (Generated), Patient Initials, Subject ID (Patient ID), Test Heart Rate (Heart
rate), Left Region Test Result 1, Left Region Test Result 3, Left Region Test Result 6, Left
Region Test Result 8, Right Region Test Result 1, Right Region Test Result 3, Right Region
Test Result 6, and Right Region Test Result 8.
Since the test descriptions for the Left and Right Regions are the same, they will have to be
extracted in separate data steps. The identifiers (testing date, Patient Initials, and Subject
ID) will be extracted for Left and Right Regions such that the Left and Right Region data can
be merged to create one dataset. Test Heart Rate will be extracted with the Left Region data
only.

Left Region Extraction
%macro Tst1lft(filename);
data Tst1lftx;
infile "(Data Location)\&filename..txt" firstobs=3 truncover
scanover;
input
@'Generated:' itstdt $100.
@'Patient Initials:' iinit $100.
@'Patient ID:' isubjid $100.
@'Heart rate:' itst1hr $100.
@'LEFT' lft $100.;
if trim(lft)='REGION RESULTS' then do;
input
@'Test Result 1:' itstrslt1 $100.
@'Test Result 3:' itstrslt3 $100.
@'Test Result 6:' itstrslt6 $100.
@'Test Result 8:' itstrslt8 $100.;
end;
call symput('subjid',strip(trim(isubjid)));
run;
Call this macro with the filename as the parameter:
%Tst1rgt(Subj 0101 LB Test 1);
Resulting in the following output:

Figure 10. Test 1 Left Region Result

Right Region Extraction macro
%macro Tst1rgt(filename,subjid);
data Tst1rgtx;
infile "(Data Location)\&filename..txt" firstobs=3 truncover

scanover;
input
@'Generated:' itstdt $100.
@'Patient Initials:' iinit $100.
@'Patient ID:' isubjid $100.
@'RIGHT' rgt $100.;
if trim(rgt)='REGION RESULTS' then do;
input
@'Test Result 1:' itstrslt1
@'Test Result 3:' itstrslt3
@'Test Result 6:' itstrslt6
@'Test Result 8:' itstrslt8
end;
call symput('subjid',strip(trim(isubjid)));

$100.
$100.
$100.
$100.;

run;
data Tst1rgt&subjid. (drop=itstdt iinit isubjid rgt itstrslt1 itstrslt3
itstrslt6 itstrslt8);
retain subjid;
length subjid $25. init $3.;
format tstdt date9.;
set Tst1rgtx;
tstdt=input(scan(itstdt,1,' '),mmddyy10.);
subjid=strip(trim(isubjid));
init=strip(trim(iinit));
rgttst1rslt1=input(scan(itstrslt1,1,' '),best12.);
rgttst1rslt3=input(scan(itstrslt3,1,' '),best12.);
rgttst1rslt6=input(scan(itstrslt6,1,' '),best12.);
rgttst1rslt8=input(scan(itstrslt8,1,' '),best12.);
run;
proc sort data=Tst1rgt&subjid; by subjid init tstdt; run;
%mend;
Call this macro with the filename as the parameter:
%Tst1rgt(Subj 0101 LB Test 1);
Resulting in the following output:

Figure 11. Test 1 Right Region Result

Create final dataset
All extracted data for the subject can now be merged into one dataset. Again, defined as a
macro where the parameter is the Subject ID, in order to more easily merge all data per
subject.
%macro mergeDData(subjid);
data DeviceTextData&subjid;
merge tst1lft&subjid.
tst1rgt&subjid.
tst2&subjid.;

by subjid init tstdt;
label subjid='Subject ID'
init='Subject Initials'
tstdt='Test Date'
tst1hr='Subject Test 1 Heart Rate'
lfttst1rslt1='Left Region Test 1 Result 1'
lfttst1rslt3='Left Region Test 1 Result 3'
lfttst1rslt6='Left Region Test 1 Result 6'
lfttst1rslt8='Left Region Test 1 Result 8'
rgttst1rslt1='Right Region Test 1 Result 1'
rgttst1rslt3='Right Region Test 1 Result 3'
rgttst1rslt6='Right Region Test 1 Result 6'
rgttst1rslt8='Right Region Test 1 Result 8'
tst2hr='Subject Test 2 Heart Rate'
tst2rslt6='Test 2 Result 6'
tst2rslt7='Test 2 Result 7';
run;
%mend mergeDData;
Call this macro with the subjid as the parameter:
%mergeDData(0101);
Resulting in the following output:

Figure 12. Final dataset for subject

CONCLUSION
In some cases, it is necessary to collect subject data one file per subject per result. Storing
each subject’s data using the same naming convention for the files and folders in one folder
per subject allows for easier import into one dataset. One of the macros presented above
can be used to import this data.
It is often practice to transcribe the results needed for research from device results in txt
format from the text reports onto a CRF, where the data on the CRF are later entered into a
database. Using SAS® to read the desired data values directly from the text report into a
dataset can eliminate the need for this process.
This process can easily be adjusted to retain units for value conversions to one standard unit
in case results provided are presented in different units.

